TRAINING ON COMPUTABLE
GENERAL EQUILIBRIUM MODELLING

MODEL AUTA

A CLOSED ECONOMY WITHOUT GOVERNMENT

This pedagogical material was developed by Véronique Robichaud and accompanies Lesson 5. The model used is an adaptation of the model AUTA presented in: Decaluwé, B., A. Martens and L. Savard (2001), « La politique économique du développement et les modèles d'équilibre général calculable. Une introduction », Montréal, Presses de l'Université de Montréal, 524 p. Since the SAM has been changed, the simulation results cannot be compared.

Table of content

Hypotheses

- Model of a closed economy without government
- Three industries / commodities:
  - Agriculture;
  - Manufacturing;
  - Services.
- Two factors of production:
  - Labour (mobile across sectors)
  - Capital (fixed by sector)
- Two categories of households:
  - Salaried households;
  - Capitalists.

Sets

Industries and commodities

$i, j \in I = \{AGR, MAN, SER\}$

(AGR: agriculture, MAN: manufacturing, SER: services)

Sub-set of industries and commodities excluding services:

$bns \in BNS \subset I = \{AGR, MAN\}$

Households

$h \in H = \{SAL, CAP\}$

(SAL: salaried, CAP: capitalists)

Equations

Production

		Number of equations
1.	$VA_j = v_j \cdot XS_j$	3
2.	$CI_j = io_j \cdot XS_j$	3
3.	$VA_j = A_j \cdot LD_j^{\alpha_j} \cdot KD_j^{1-\alpha_j}$	3
4.	$LD_j = \frac{\alpha_j \cdot PVA_j \cdot VA_j}{W}$	3
5.	$KD_j = \frac{(1-\alpha_j) \cdot PVA_j \cdot VA_j}{R_j}$	3
6.	$DI_{i,j} = aij_{i,j} \cdot CI_j$	9

Income and savings

7.	$YH_{'SAL'} = W \cdot \sum_j LD_j$	1
8.	$YH_{'CAP'} = \lambda \cdot \sum_j R_j \cdot KD_j + DIV$	1
9.	$SH_h = \psi_h \cdot YH_h$	2
10.	$CTH_h = YH_h - SH_h$	2
11.	$YF = (1-\lambda) \cdot \sum_j R_j \cdot KD_j$	1
12.	$SF = YF - DIV$	1

Demand

13.	$C_{i,h} = \frac{\gamma_{i,h} \, CTH_h}{P_i}$	6
14.	$INV_i = \frac{\mu_i \, IT}{P_i}$	3
15.	$DIT_i = \sum_j DI_{i,j}$	3

Prices

16. $P_j = \frac{PVA_j \cdot VA_j + PCI_j \cdot CI_j}{XS_j}$ 3

17. $PCI_j = \frac{\sum_i P_i \ DI_{i,j}}{CI_j}$ 3

Equilibrium

18. $XS_{bns} = \sum_h C_{h,bns} + DIT_{bns} + INV_{bns}$ 2

19. $LS = \sum_j LD_j$ 1

20. $KS_j = KD_j$ 3

21. $IT = \sum_h SH_h + SF$ 1

Verification of the Walras' law

22. $LEON = XS_{'SER'} - \sum_h C_{h,'SER'} - DIT_{'SER'} - INV_{'SER'}$ 1

Total: 58

Variables

Volume variables (quantities)

		Number
$C_{i,h}$:	Consumption of commodity *i* by type *h* households	6
CI_j :	Total intermediate consumption of industry *j*	3
$DI_{i,j}$:	Intermediate consumption of commodity *i* in industry *j*	9
DIT_i :	Total intermediate demand for commodity *i*	3
INV_i :	Final demand of commodity *i* for investment purposes	3
KD_j :	Industry *j* demand for capital	3
KS_j :	Capital supply in industry *j*	3
LD_j :	Industry *j* demand for labour	3
LS :	Total labour supply	1
VA_j :	Value added of industry *j*	3
XS_j :	Output of industry *j*	3

Price variables

P_i :	Price of commodity *i*	3
PCI_j :	Intermediate consumption price index of industry *j*	3
PVA_j :	Price of industry *j* value added	3
R_j :	Rental rate of capital in industry *j*	3
W :	Wage rate	1

Nominal variables (values)

CTH_h :	Consumption budget of type *h* households	2
DIV :	Dividends	1
IT :	Total investment	1
SF :	Business savings	1
SH_h :	Savings of type *h* households	2
YF :	Business income	1
YH_h :	Income of type *h* households	2

Other variables

$LEON$:	Excess supply on the market for services	1
		Total: 64

Exogenous variables (closures)		Number
DIV :	Dividends	1
KS_j :	Capital supply in industry *j*	3
LS :	Total labour supply	1
P_{agr} :	Price of commodity *agr* – numéraire	1
		Total: 6

Parameters

A_j : Scale parameter (Cobb-Douglas – production function)

$aij_{i,j}$: Coefficient (Leontief - intermediate consumption)

α_j : Elasticity (Cobb-Douglas – production function)

$\gamma_{i,h}$: Share of commodity *i* in type *h* household consumption budget

io_j : Coefficient (Leontief – total intermediate consumption)

λ : Share of capital income received by capitalists

μ_i : Share of commodity *i* in total investment expenditures

ψ_h : Average propensity to save of type *h* household

v_j : Coefficient (Leontief – value added)

The social accounting matrix for AUTA

	FACTORS		AGENTS			INDUSTRIES/COMMODITIES			ACC.	TOTAL
	1.	2.	3.	4.	5.	6.	7.	8.	9.	(1 to 9)
1. Labour						300	100	200		600
2. Capital						100	150	100		350
3. Salaried households	600									600
4. Capitalists		210			70					280
5. Firms		140								140
6. Agriculture			162	21		50	150	90	27	500
7. Manufacturing			108	84		20	150	90	173	625
8. Services			270	105		30	75	120		600
9. Accumulation			60	70	70					200
Total (1 to 9)	600	350	600	280	140	500	625	600	200	

Correspondence between the SAM and the model



Schema


KD_j
R_j
LD_j
W
VA_j
PVA_j
XS_j
P_j
CI_j
PCI_j
$DI_{i,j}$
DIT_i
P_i
YH_{sal}
$C_{i,h}$
P_i
YH_{cap}
YF
$ST = IT$
INV_i
P_i

GAMS Code

```
$TITLE    MODEL AUTA
$STITLE   AUTARKY WITHOUT GOVERNMENT

* Model of a closed economy without government producing 3 goods using
* 2 factors owned by 2 types of households.

* October 2016

* Acknowledgement

* This GAMS code was prepared by Veronique Robichaud. This basic CGE model is
* taken from Decaluwe, B., A. Martens and L. Savard (2001), " La politique
* economique du developpement et les modeles d'equilibre general calculable.
* Une introduction ", Montreal, Presses de l'Université de Montreal, 524 p.

* CALIBRATION

* Sets definition

SETS I Industries and commodities
/ AGR agriculture
  MAN manufacturing
  SER services /

BNS(I) Goods
/ AGR agriculture
  MAN manufacturing /

H Households
/ SAL labour endowed households
  CAP capital endowed households /

ALIAS (i,j)
;

* Parameters definition

PARAMETERS

 A(j)            Scale parameter (Cobb-Douglas - production function)
 aij(i,j)        Coefficient (Leontief - intermediate consumption)
 alpha(j)        Elasticity (Cobb-Douglas - production function)
 gamma(i,h)      Share of commodity i in type h household consumption budget
 io(j)           Coefficient (Leontief - total intermediate consumption)
 lambda          Share of capital income received by capitalists
 mu(i)           Share of commodity i in total investment expenditures
 psi(h)          Average propensity to save of type h household
 v(j)            Coefficient (Leontief - value added)
```

```
* Definition of variables for the base year
* Volume variables (quantities)
 CO(i,h)          Consumption of commodity i by type h households
 CIO(j)           Total intermediate consumption of industry j
 DIO(i,j)         Intermediate consumption of commodity i in industry j
 DITO(i)          Total intermediate demand for commodity i
 INVO(i)          Final demand of commodity i for investment purposes
 KDO(j)           Industry j demand for capital
 KSO(j)           Capital supply in industry j
 LDO(j)           Industry j demand for labour
 LSO              Total labour supply
 VAO(j)           Value added of industry j
 XSO(j)           Output of industry j

* Prices
 PO(i)            Price of commodity i
 PCIO(j)          Intermediate consumption price index of industry j
 PVAO(j)          Price of industry j value added
 RO(j)            Rental rate of capital in industry j
 WO               Wage rate

* Nominal variables (values)
 CTHO(h)          Consumption budget of type h households
 DIVO             Dividends
 ITO              Total investment
 SFO              Business savings
 SHO(h)           Savings of type h households
 YFO              Business income
 YHO(h)           Income of type h households
;


* Initial data

TABLE SAM(*,*) Social accounting matrix

          LD      KD      SAL      CAP        F      AGR      MAN      SER      ACC
LD                                                  300      100      200
KD                                                  100      150      100
SAL       600
CAP               210                       70
F                 140
AGR                       162       21               50      150       90       27
MAN                       108       84               20      150       90      173
SER                       270      105               30       75      120
ACC                        60       70       70
TOT       600     350     600      280      140      500      625      600      200
;
```

```
* Assignment of variables
 DIVO            = SAM('CAP','F');
 ITO             = SAM('TOT','ACC');
 SFO             = SAM('ACC','F');
 SHO(h)          = SAM('ACC',h);
 YFO             = SAM('TOT','F');
 YHO(h)          = SAM('TOT',h);
** For some variables in volume, we first assign a temporary value from
** the SAM. Later on, we will divide this value by the corresponding price.
 XSO(j)          = SAM('TOT',j);
 CO(i,h)         = SAM(i,h);
 DIO(i,j)        = SAM(i,j);
 INVO(i)         = SAM(i,'ACC');
 KDO(j)          = SAM('KD',j);
 LDO(j)          = SAM('LD',j);

* Prices
 WO              = 1;
 RO(j)           = 1;
 PO(i)           = 1;

* Computation of variables in volume
 LDO(j)          = LDO(j)/WO;
 KDO(j)          = KDO(j)/RO(j);
 XSO(j)          = XSO(j)/PO(j);
 CO(i,h)         = CO(i,h)/PO(i);
 INVO(i)         = INVO(i)/PO(i);
 DIO(i,j)        = DIO(i,j)/PO(i);

* Calibration of other variables
 LSO             = SUM[i,LDO(i)];
 KSO(j)          = KDO(j);
 VAO(j)          = LDO(j)+KDO(j);
 PVAO(j)         = {WO*LDO(j)+RO(j)*KDO(j)}/VAO(j);
 DITO(i)         = SUM[j,DIO(i,j)];
 CIO(j)          = SUM[i,DIO(i,j)];
 PCIO(j)         = SUM[i,PO(i)*DIO(i,j)]/CIO(j);
 CTHO(h)         = YHO(h)-SHO(h);

* Calibration of parameters
* Production (Cobb-Douglas and Leontief)
 alpha(j)        = WO*LDO(j)/{PVAO(j)*VAO(j)};
 A(j)            = VAO(j)/{LDO(j)**alpha(j)*KDO(j)**(1-alpha(j))};
 v(j)            = VAO(j)/XSO(j);
 io(j)           = CIO(j)/XSO(j);
 aij(i,j)        = DIO(i,j)/CIO(j);
```

```
* Distribution parameters
 gamma(i,h)       = PO(i)*CO(i,h)/CTHO(h);
 lambda           = {YHO('cap')-DIVO}/SUM[j,RO(j)*KDO(j)];
 mu(i)            = PO(i)*INVO(i)/ITO;
 psi(h)           = SHO(h)/YHO(h);

* Parameters to be displayed in the output file
DISPLAY A, alpha, io, v, aij, gamma, psi, mu, lambda;

* MODEL

* Definition of variables

VARIABLES

* Volume variables (quantities)
 C(i,h)           Consumption of commodity i by type h households
 CI(j)            Total intermediate consumption of industry j
 DI(i,j)          Intermediate consumption of commodity i in industry j
 DIT(i)           Total intermediate demand for commodity i
 INV(i)           Final demand of commodity i for investment purposes
 KD(j)            Industry j demand for capital
 KS(j)            Capital supply in industry j
 LD(j)            Industry j demand for labour
 LS               Total labour supply
 VA(j)            Value added of industry j
 XS(j)            Output of industry j

* Prices
 P(i)             Price of commodity i
 PCI(j)           Intermediate consumption price index of industry j
 PVA(j)           Price of industry j value added
 R(j)             Rental rate of capital in industry j
 W                Wage rate

* Nominal variables (values)
 CTH(h)           Consumption budget of type h households
 DIV              Dividends
 IT               Total investment
 SF               Business savings
 SH(h)            Savings of type h households
 YF               Business income
 YH(h)            Income of type h households

* Other variable
 LEON             Excess supply on the market for services
;
```

```
* Definition of equations

* Production
EQUATIONS
 XSEQ(j)          Value added demand in industry j (Leontief)
 CIEQ(j)          Total intermediate consumption demand in industry j (Leontief)
 VAEQ(j)          Cobb-Douglas between labour and capital
 LDEQ(j)          Demand for labour by industry j
 KDEQ(j)          Demand for capital by industry j
 DIEQ(i,j)        Intermediate consumption of commodity i by sector j
;

 XSEQ(j)..        VA(j) =e= v(j)*XS(j);

 CIEQ(j)..        CI(j) =e= io(j)*XS(j);

 VAEQ(j)..        VA(j) =e= A(j)*LD(j)**alpha(j)*KD(j)**(1-alpha(j)) ;

 LDEQ(j)..        W*LD(j) =e= alpha(j)*PVA(j)*VA(j);

 KDEQ(j)..        R(j)*KD(j) =e= (1-alpha(j))*PVA(j)*VA(j);

 DIEQ(i,j)..      DI(i,j) =e= aij(i,j)*CI(j);

* Income and savings
EQUATIONS
 YHSEQ            Household income (workers)
 YHCEQ            Household income (capitalists)
 SHEQ(h)          Household h savings
 CTHEQ(h)         Consumption budget
 YFEQ             Firms income
 SFEQ             Firms savings
;

 YHSEQ..          YH('sal') =e= W*SUM[j,LD(j)];

 YHCEQ..          YH('cap') =e= lambda*SUM[j,R(j)*KD(j)]+DIV;

 SHEQ(h)..        SH(h) =e= psi(h)*YH(h);

 CTHEQ(h)..       CTH(h) =e= YH(h)-SH(h);

 YFEQ..           YF =e= (1-lambda)*SUM[j,R(j)*KD(j)];

 SFEQ..           SF =e= YF-DIV;
```

```
* Demand
EQUATIONS
 CEQ(i,h)         Household h consumption of commodity i
 INVEQ(i)         Investment in commodity i
 DITEQ(i)         Intermediate demand for commodity i
;

 CEQ(i,h)..       P(i)*C(i,h) =e= gamma(i,h)*CTH(h);

 INVEQ(i)..       P(i)*INV(i) =e= mu(i)*IT;

 DITEQ(i)..       DIT(i) =e= SUM[j,DI(i,j)];

* Prices
EQUATIONS
 PCIEQ(j)         Intermediate consumption price index
 CPEQ(j)          Production costs for sector j
;

 PCIEQ(j)..       PCI(j)*CI(j) =e= SUM[i,P(i)*DI(i,j)];

 CPEQ(j)..        P(j)*XS(j) =e= PVA(j)*VA(j)+PCI(j)*CI(j);


* Équilibre
EQUATIONS
 PEQ(bns)         Domestic absorption
 WEQ              Labour market equilibrium
 REQ(j)           Capital market equilibrium
 ITEQ             Investment-savings equilibrium
;
 PEQ(bns)..       XS(bns) =e= SUM[h,C(bns,h)]+DIT(bns)+INV(bns);

 WEQ..            LS =e= SUM[j,LD(j)];

 REQ(j)..         KS(j) =e= KD(j);

 ITEQ..           IT =e= SUM[h,SH(h)]+SF;

* Other
EQUATIONS
 WALRAS           Verification of the Walras law
;

WALRAS..          LEON =e= XS('ser')-SUM(h,C('ser',h))-DIT('ser')-INV('ser');
```

```
* Initialisation of variables

 C.l(i,h)         = CO(i,h);
 CI.l(j)          = CIO(j);
 CTH.l(h)         = CTHO(h);
 DI.l(i,j)        = DIO(i,j);
 DIT.l(i)         = DITO(i);
 DIV.l            = DIVO;
 INV.l(i)         = INVO(i);
 IT.l             = ITO;
 KD.l(j)          = KDO(j);
 KS.l(j)          = KSO(j);
 LD.l(j)          = LDO(j);
 LS.l             = LSO;
 P.l(i)           = PO(i);
 PCI.l(j)         = PCIO(j);
 PVA.l(j)         = PVAO(j);
 R.l(j)           = RO(j);
 SF.l             = SFO;
 SH.l(h)          = SHO(h);
 VA.l(j)          = VAO(j);
 W.l              = WO;
 XS.l(j)          = XSO(j);
 YF.l             = YFO;
 YH.l(h)          = YHO(h);

* Closures

* P(AGR) is the numéraire
 P.fx('agr')      = PO('agr');

* Capital is sector specific
 KS.fx(j)         = KSO(j);

* Total labour suply is fixed
 LS.fx            = LSO;

* Dividends are exogenous
 DIV.fx           = DIVO;

* Model execution

MODEL AUTA Autarky without government /ALL/;
AUTA.HOLDFIXED=1;
SOLVE AUTA USING CNS;
```

Value of parameters

Definition	Symbol	Value
Scale parameter – Cobb-Douglas	A_{agr}	1.755
	A_{man}	1.960
	A_{ser}	1.890
Intermediate consumption coefficient – Leontief	$aij_{agr,agr}$	0.500
	$aij_{man,agr}$	0.200
	$aij_{ser,agr}$	0.300
	$aij_{agr,man}$	0.400
	$aij_{man,man}$	0.400
	$aij_{ser,man}$	0.200
	$aij_{agr,ser}$	0.300
	$aij_{man,ser}$	0.300
	$aij_{ser,ser}$	0.400
Elasticity – Cobb-Douglas	α_{agr}	0.750
	α_{man}	0.400
	α_{ser}	0.667
Share of commodity in the consumption budget of salaried households	$\gamma_{agr,sal}$	0.300
	$\gamma_{man,sal}$	0.200
	$\gamma_{ser,sal}$	0.500
Share of commodity in the consumption budget of capitalists households	$\gamma_{agr,cap}$	0.100
	$\gamma_{man,cap}$	0.400
	$\gamma_{ser,cap}$	0.500
Total intermediate consumption coefficient – Leontief	io_{agr}	0.200
	io_{man}	0.600
	io_{ser}	0.500
Share of capital income to capitalists	λ	0.600
Share of commodity in total investment expenditures	μ_{agr}	0.135
	μ_{man}	0.865
	μ_{serl}	
Propensity to save	ψ_{sal}	0.100
	ψ_{cap}	0.250
Value-added coefficient – Leontief	v_{agr}	0.800
	v_{man}	0.400
	v_{ser}	0.500

Simulations

SIM1: Impact of a 10% increase of capital supply in services

Definition	Symbol	Initial value	Simulation	Variation (%)
PRICES				
• wage rate	W	1	1.000	-0.019
Rental rate of capital				
• agriculture	R_{agr}	1	1.008	0.753
• manufacturing	R_{man}	1	1.012	1.223
• services	R_{ser}	1	0.893	-10.726
Price of value added				
• agriculture	PVA_{agr}	1	1.002	0.174
• manufacturing	PVA_{man}	1	1.007	0.724
• services	PVA_{ser}	1	0.963	-3.723
Intermediate consumption price index				
• agriculture	PCI_{agr}	1	0.993	-0.694
• manufacturing	PCI_{man}	1	0.995	-0.459
• services	PCI_{ser}	1	0.991	-0.925
Price of commodity				
• **agriculture (numéraire)**	$\boldsymbol{P_{agr}}$	1	1.000	-
• manufacturing	P_{man}	1	1.000	0.014
• services	P_{ser}	1	0.977	-2.324
PRODUCTION AND FACTORS				
Output				
• agriculture	XS_{agr}	500	502.894	0.579
• manufacturing	XS_{man}	625	628.094	0.495
• services	XS_{ser}	600	611.998	2.000
Value added				
• agriculture	VA_{agr}	400	402.315	0.579
• manufacturing	VA_{man}	250	251.237	0.495
• services	VA_{ser}	300	305.999	2.000
Labour				
• agriculture	LD_{agr}	300	302.317	0.772
• manufacturing	LD_{man}	100	101.242	1.242
• services	LD_{ser}	200	196.441	-1.780
• total	LS	600	600	-
Capital				
• agriculture	KD_{agr}	100	100.000	-
• manufacturing	KD_{man}	150	150.000	-
• services	KD_{ser}	100	110.000	10.000

Total intermediate consumption				
• agriculture	CI_{agr}	100	100.579	0.579
• manufacturing	CI_{man}	375	376.856	0.495
• services	CI_{ser}	300	305.999	2.000
Intermediate consumption				
• agriculture	$DI_{agr.agr}$	50	50.289	0.579
	$DI_{man.agr}$	20	20.116	0.579
	$DI_{ser.agr}$	30	30.174	0.579
• manufacturing	$DI_{agr.man}$	150	150.742	0.495
	$DI_{man.man}$	150	150.742	0.495
	$DI_{ser.man}$	75	75.371	0.495
• services	$DI_{agr.ser}$	90	91.800	2.000
	$DI_{man.ser}$	90	91.800	2.000
	$DI_{ser.ser}$	120	122.400	2.000
INCOME AND SAVINGS				
Income				
• workers	YH_{sal}	600	599.887	-0.019
• capitalists	YH_{cap}	280	280.474	0.169
• firms	YF	140	140.316	0.226
Savings				
• workers	SH_{sal}	60	59.989	-0.019
• capitalists	SH_{cap}	70	70.118	0.169
• firms	SF	70	70.316	0.451
DEMAND				
Salaried households – consumption				
• agriculture	$C_{agr.sal}$	162	161.969	-0.019
• manufacturing	$C_{man.sal}$	108	107.964	-0.033
• services	$C_{ser.sal}$	270	276.373	2.360
• total	CTH_{sal}	540	539.898	-0.019
Capitalists – consumption				
• agriculture	$C_{agr.cap}$	21	21.036	0.169
• manufacturing	$C_{man.cap}$	84	84.130	0.155
• services	$C_{ser.cap}$	105	107.681	2.553
• total	CTH_{cap}	210	210.355	0.169
Intermediate demand				
• agriculture	DIT_{agr}	290	292.832	0.976
• manufacturing	DIT_{man}	260	262.658	1.022
• services	DIT_{ser}	225	227.944	1.309
Demand for investment purposes				
• agriculture	INV_{agr}	27	27.057	0.211
• manufacturing	INV_{man}	173	173.341	0.197
• total	IT	200	200.423	0.211

SIM2: Impact of a 10% of labour supply

Definition	Symbol	Initial value	Simulation	Variation (%)
PRICES				
• wage rate	W	1	0.976	-2.361
Rental rate of capital				
• agriculture	R_{agr}	1	1.064	6.429
• manufacturing	R_{man}	1	1.117	11.743
• services	R_{ser}	1	1.067	6.692
Price of value added				
• agriculture	PVA_{agr}	1	0.998	-0.234
• manufacturing	PVA_{man}	1	1.059	5.872
• services	PVA_{ser}	1	1.006	0.568
Intermediate consumption price index				
• agriculture	PCI_{agr}	1	1.009	0.937
• manufacturing	PCI_{man}	1	1.015	1.490
• services	PCI_{ser}	1	1.014	1.358
Price of commodity				
• **agriculture (numéraire)**	$\boldsymbol{P_{agr}}$	1	1.000	-
• manufacturing	P_{man}	1	1.032	3.242
• services	P_{ser}	1	1.010	0.963
PRODUCTION AND FACTORS				
Output				
• agriculture	XS_{agr}	500	533.395	6.679
• manufacturing	XS_{man}	625	659.658	5.545
• services	XS_{ser}	600	636.540	6.090
Value added				
• agriculture	VA_{agr}	400	426.716	6.679
• manufacturing	VA_{man}	250	263.863	5.545
• services	VA_{ser}	300	318.270	6.090
Labour				
• agriculture	LD_{agr}	300	327.009	9.003
• manufacturing	LD_{man}	100	114.445	14.445
• services	LD_{ser}	200	218.546	9.273
• total	LS	600	660	10.000
Capital				
• agriculture	KD_{agr}	100	100.000	-
• manufacturing	KD_{man}	150	150.000	-
• services	KD_{ser}	100	100.000	-

Total intermediate consumption				
• agriculture	CI_{agr}	100	106.679	6.679
• manufacturing	CI_{man}	375	395.795	5.545
• services	CI_{ser}	300	318.270	6.090
Intermediate consumption				
• agriculture	$DI_{agr.agr}$	50	53.340	6.679
	$DI_{man.agr}$	20	21.336	6.679
	$DI_{ser.agr}$	30	32.004	6.679
• manufacturing	$DI_{agr.man}$	150	158.318	5.545
	$DI_{man.man}$	150	158.318	5.545
	$DI_{ser.man}$	75	79.159	5.545
• services	$DI_{agr.ser}$	90	95.481	6.090
	$DI_{man.ser}$	90	95.481	6.090
	$DI_{ser.ser}$	120	127.308	6.090
INCOME AND SAVINGS				
Income				
• workers	YH_{sal}	600	644.415	7.402
• capitalists	YH_{cap}	280	298.441	6.586
• firms	YF	140	152.294	8.782
Savings				
• workers	SH_{sal}	60	64.441	7.402
• capitalists	SH_{cap}	70	74.610	6.586
• firms	SF	70	82.294	17.563
DEMAND				
Salaried households – consumption				
• agriculture	$C_{agr.sal}$	162	173.992	7.402
• manufacturing	$C_{man.sal}$	108	112.352	4.029
• services	$C_{ser.sal}$	270	287.221	6.378
• total	CTH_{sal}	540	579.973	7.402
Capitalists – consumption				
• agriculture	$C_{agr.cap}$	21	22.383	6.586
• manufacturing	$C_{man.cap}$	84	86.721	3.239
• services	$C_{ser.cap}$	105	110.848	5.570
• total	CTH_{cap}	210	223.831	6.586
Intermediate demand				
• agriculture	DIT_{agr}	290	307.139	5.910
• manufacturing	DIT_{man}	260	275.135	5.821
• services	DIT_{ser}	225	238.471	5.987
Demand for investment purposes				
• agriculture	INV_{agr}	27	29.882	10.673
• manufacturing	INV_{man}	173	185.451	7.197
• total	IT	200	221.346	10.673